On the Calculation of International Roughness Index from Longitudinal Road Profile

MICHAEL W. SAYERS

The international roughness index (IRI) was established in 1986 by the World Bank and based on earlier work performed for NCHRP. IRI is calculated from a measured longitudinal road profile by accumulating the output from a quarter-car model and dividing by the profile length to yield a summary roughness index with units of slope. Although IRI is used widely, there is no single, short reference document that describes what it is and how it is calculated. Instead, the critical information is spread over several large reports. A short, self-contained reference that defines IRI is provided, along with all the information needed to compute it from longitudinal road profile measurements. The development of the IRI is reviewed, the mathematical definition is presented, an algorithm for calculating IRI is derived, the performance of the algorithm is analyzed, tested Fortran source code for computing IRI is presented, and problems with IRI (and profile measurement in general) that have emerged since 1986 are identified.

The international roughness index (IRI) evolved over many years, in three stages:

1. Quarter-car simulation on high-speed profilers. Routine analysis of road profiles began shortly after the General Motors (GM) profilometer was developed in the late 1960s by Spangler and Kelley (*1*). Like high-speed profilers today, it could measure true profile over a range of wavelengths affecting vehicle vibrations. One of the first research applications for this type of system combined measured road profiles with a quarter-car computer model that replicated the Bureau of Public Roads (BPR) Roughometer, a one-wheeled trailer with a road meter (*2,3*). GM licensed K.J. Law, Inc. to market the device commercially and continue its development. A commercial version was soon available that included a quarter-car analysis to summarize roughness of the measured profiles. Users of early K.J. Law profilometers could choose between two quarter-car data sets: one for the BPR Roughometer and one for a 1968 Chevrolet Impala (*4*).

2. NCHRP research and the Golden Car. In the late 1970s, NCHRP sponsored a study of response-type road roughness measuring systems such as the BPR Roughometer and vehicles equipped with Mays ride meters. The results were published in *NCHRP Report 228* (*5*). An objective of the study was to develop calibration methods for the response-type systems. The researchers, Gillespie and Sayers, concluded that the only valid method was *calibration by correlation* against a defined roughness index. Considerable research was performed using simulations and experiments to compare alternative reference roughness indexes. The candidate analyses included vehicle simulation with 10 alternative sets of parameters. The best correlation was obtained by using a vehicle simulation with a set of parameter values that is often called the *Golden Car*. (The name is based on the concept of a golden reference instrument kept in a vault and used to calibrate other instruments.)

Some researchers and users assume that the Golden Car parameters describe an average American passenger car, circa 1978. This is not the case. Spring rates were selected to match the two major resonant frequencies (body and axle bounce), but damping in the Golden Car is much higher than in most cars and trucks. The high damping was chosen because the computer study showed that it improved correlation with a wide variety of response-type systems.

The NCHRP study provided a standard quarter-car model, and users of K.J. Law profilometers soon had access to an analysis called *Mays simulation*, which used the Golden Car data set.

3. World Bank development of IRI. In 1982 the World Bank initiated a correlation experiment in Brazil called the International Road Roughness Experiment (IRRE) to establish correlation and a calibration standard for roughness measurements (*6*). In processing the data, it became clear that nearly all roughness-measuring instruments in use throughout the world were capable of producing measures on the same scale, if that scale had been selected suitably. Accordingly, an objective was added to the research program: develop the IRI.

The main criteria in designing the IRI were that it be relevant, transportable, and stable with time. To ensure transportability, it had to be measurable with a wide range of equipment, including response-type systems. To be stable with time, it had to be defined as a mathematical transform of a measured profile. Many roughness definitions were applied to the large amount of test data obtained in the IRRE. The Golden Car simulation from the NCHRP project was one of the candidate references considered, under the condition that a standard simulation speed would be needed to use it for the IRI. After processing the IRRE data, the best correlations between a profile index and the response-type systems were found with two vehicle simulations based on the Golden Car parameters: a quarter-car and a half-car. Both gave essentially the same level of correlation. The quarter-car was selected for the IRI because it could be used with *all* profiling methods that were in use at that time. The consensus of the researchers and participants is that the standard speed should be 80 km/hr (49.7 mph) because at that simulated speed, the IRI is sensitive to the same profile wavelengths that cause vehicle vibrations in normal highway use.

The research findings were highly encouraging and led the World Bank to publish guidelines for conducting and calibrating rough-

University of Michigan Transportation Research Institute, 2901 Baxter Road, Ann Arbor, Mich. 48109.

ness measurements. The researchers (Sayers, Gillespie, Queiroz, and Paterson) prepared instructions for using various types of equipment to measure IRI (*7*). The guidelines also include computer code for calculating IRI from profile. A companion report (*6*) described the IRRE, using many analytical comparisons of algorithms and some sensitivity analyses. In 1990 FHWA required the IRI as the standard reference for reporting roughness in the Highway Performance Monitoring System (HPMS) (*8*).

OBJECTIVES OF PAPER

The main objective of this paper is to provide a self-contained description of IRI, including its definition and an algorithm for its calculation. Well-tested Fortran source code for computing IRI is provided for the benefit of those persons developing software to analyze profile measurements. Previously unpublished background theory is provided about how the algorithm works, in an attempt to dispel some errors and misconceptions about IRI that have appeared in the past decade. The paper also describes some unresolved issues that have come to light after years of measuring IRI with a variety of profiling equipment and methods.

DEFINITION OF IRI

The following points fully define the IRI concept; implications of these points are discussed later:

1. IRI is computed from a single longitudinal profile. The sample interval should be no larger than 300 mm for accurate calculations. The required resolution depends on the roughness level, with finer resolution being needed for smooth roads. A resolution of 0.5 mm is suitable for all conditions.
2. The profile is assumed to have a constant slope between sampled elevation points.
3. The profile is smoothed with a moving average whose base length is 250 mm.
4. The smoothed profile is filtered using a quarter-car simulation, with specific parameter values (Golden Car), at a simulated speed of 80 km/hr (49.7 mph).
5. The simulated suspension motion is linearly accumulated and divided by the length of the profile to yield IRI. Thus, IRI has units of slope, such as inches per mile or meters per kilometer.

IRI Input

Number of Profiles

The IRI is defined as a property of a single wheel-track profile. For systems that measure several profiles simultaneously, it is calculated independently for each.

An alternative analysis is sometimes done when two profiles are measured at the same time. The profiles are averaged, point by point, and then processed using the IRI algorithm. This form of analysis is called a half-car simulation, and it is not the same as IRI. Similarities and differences between the half-car roughness index (HRI) and IRI are discussed elsewhere (*9*).

IRI was defined for a single wheel-track profile because (*a*) many profiler instruments can measure only one profile at a time; (*b*) for labor-intensive methods such as rod and level or DipStick, the cost of measuring two profiles is twice that of measuring one; and (*c*) in the research programs mentioned earlier, the correlation between IRI and HRI was so high that the two were statistically interchangeable.

Digital Sampling

Profile analysis is nearly always performed numerically with a digital computer. The profile is sampled to obtain a sequence of elevation numbers, where each number corresponds to a different location along the profile. The longitudinal separation between samples is a constant, Δ, which depends on the type of equipment used to obtain the profile and possibly settings made by the operator.

Filters such as the IRI quarter car are defined for a continuous profile, which means that there is an underlying assumption about what the profile does between samples. Figure 1 shows several possible methods for interpolating between sampled values:

1. Zero slope between points implies a discontinuity in elevation at each sample location,
2. Linear interpolation between points implies a constant slope, and
3. Quadratic interpolation maintains continuity in both elevation and slope through the sample values.

A computer study was done in the preparation for the IRRE to determine which of the three interpolation methods gave the best accuracy for various values of Δ. All three methods give the same IRI values when Δ is very small, on the order of 50 mm (2 in.) or smaller. However, for larger Δ-values, results calculated using Option 1 were too high, results calculated with Option 3 were too low, and results calculated with Option 2 were reasonably accurate. For larger sample intervals, the results were too low even with Option 2. These results lead to the assumption, built into the IRI, that the profile between sampled measures is a straight line connecting the points. Limits were set on the sample interval: 300 mm for accurate measures and 600 mm for less accurate measures with some bias.

IRI Filter

The IRI includes two distinct filters: a moving average and a quarter-car model.

Moving Average

The moving average was included for two reasons: (*a*) to simulate the enveloping behavior of pneumatic tires on highway vehicles,



FIGURE 1 Methods for interpolating between profile samples: *left*, zero slope—hold previous value; *middle*, constant slope—linear interpolation; *right*, continuous slope—quadratic interpolation.

and (*b*) to reduce the sensitivity of the IRI algorithm to the sample interval, Δ. For a profile that has been sampled at Δ, a moving average smoothing filter is defined by the summation

$$h_{ps}(i) = \frac{1}{k} \sum_{j=i}^{i+k-1} h_p(j) \tag{1}$$

$$k = \max[1, \text{nint}(L_B/\Delta)] \tag{2}$$

where

- h_p = profile height,
- h_{ps} = smoothed profile height,
- max = maximum of two arguments,
- nint = nearest integer, and
- L_B = moving average base length, 250 mm.

For example, if the sample interval is Δ = 150 mm (6 in.), the ratio (L_B/Δ) is 1.67, which is rounded to 2. The number 2 is larger than 1, so k is set to 2.

Digital profilometers made by K.J. Law, Inc. have often been used in American research programs. The data acquisition software in the digital K.J. Law systems has always included a moving average of 300 mm. The difference between a 250-mm and 300-mm moving average on IRI is negligible. Thus, the moving average part of the IRI results in a high degree of compatibility with Law profilometers. However, it is important that this averaging not be performed a second time when IRI is calculated from profiles measured with Law systems. (Read the comments in the IRI code presented later in Figure 6.).

Quarter Car

Figure 2 shows the quarter-car model in the IRI. It includes the major dynamic effects that determine how roughness causes vibrations in a road vehicle. The masses, springs, and dampers are defined by the following parameters:



FIGURE 2 Quarter-car model.

- c_s = suspension damping rate
- k_s = suspension spring rate
- k_t = tire spring rate
- m_s = sprung mass (portion of vehicle body mass supported by one wheel)
- m_u = unsprung mass (mass of wheel, tire, and half of axle/suspension)

To simplify the equations, the parameters are normalized by the sprung mass, m_s. The following values for the normalized parameters define the *Golden Car* data set:

$$\begin{aligned} c &= c_s/m_s = 6.0 \\ k_1 &= k_t/m_s = 653 \\ k_2 &= k_s/m_s = 63.3 \\ \mu &= m_u/m_s = 0.15 \end{aligned} \tag{3}$$

The quarter-car model is described by four first-order ordinary differential equations that can be written in matrix form:

$$\dot{\mathbf{x}} = \mathbf{A}\mathbf{x} + \mathbf{B}\, h_{ps} \tag{4}$$

where the **x**, **A**, and **B** arrays are defined as follows:

$$\mathbf{x} = [z_s,\ \dot{z}_s,\ z_u,\ \dot{z}_u]^T$$

$$\mathbf{A} = \begin{bmatrix} 1 & 0 & 0 & 0 \\ -k_2 & -c & k_2 & c \\ 0 & 0 & 1 & 0 \\ \frac{k_2}{\mu} & \frac{c}{\mu} & -\frac{k_1 + k_2}{\mu} & -\frac{c}{\mu} \end{bmatrix} \tag{5}$$

$$\mathbf{B} = [0, 0, 0, k_1/\mu]^T$$

where

- h_{ps} = smoothed profile elevation,
- z_s = height (vertical coordinate) of sprung mass,
- z_u = height (vertical coordinate) of unsprung mass, and
- **x** = array of state variables (variables that, together, completely describe state of simulated system).

Time derivatives are indicated with a dot (e.g., $\dot{z}_s$). Time is related to longitudinal distance by the simulated speed of the vehicle

$$t = x/V \tag{6}$$

where x is longitudinal distance and V is the simulated forward speed. V is defined as 80 km/hr (49.7 mph) for the IRI. The units of V should be length/second, where the units of length match those of x (e.g., $V = 22.22\bar{2}$ m/sec = 72.90755 ft/sec).

IRI Accumulator

The IRI is an accumulation of the simulated motion between the sprung and unsprung masses in the quarter-car model, normalized by the length L, of the profile:

$$\text{IRI} = \frac{1}{L} \int_0^{L/V} \left|\dot{z}_s - \dot{z}_u\right| dt \tag{7}$$

The vehicle response variables oscillate about 0 and have 0 as an average value. The absolute value in equation 7 is needed to obtain a non-zero average. Another method that might have been selected, but was not, is root mean square (*RMS*). *RMS* averaging is less convenient to use because it is nonlinear with respect to absolute amplitude, whereas equation 7 is linear. For example, if the IRI of a 0.5-mi section is 100 in/mi, and the IRI for the next 0.5-mi section is 200 in./mi, the IRI for the entire mile is the simple linear average: 150 in./mi. In contrast, the *RMS* average would be 158. Another reason for using linear averaging was to match the behavior of existing road meters.

Initialization

To solve differential equations such as equation 4, one must know or estimate the values of the state variables at the starting time. The response obtained over a profile includes a response of the transition from the initial values to the profile-induced response. The effect of the initialization diminishes as the simulated car covers more of the profile. At the IRI simulation speed of 80 km/hr, the initialization influences the quarter-car response for about 20 m. The most accurate way to deal with the initialization is to measure the profile for 20 m or so before the site and start the simulation there. Then, at the start of the test site, begin the IRI accumulation (equation 7).

A computer study was made to find an initialization method that minimizes errors in the first 20 m (*6*). It led to the recommendations that, initially, z_s and z_u should be set to match the height of the first profile point and that $\dot{z}_s$ and $\dot{z}_u$ should be set to match the average change in profile height per second, at the simulation speed of 80 km/hr, over the first 11 m of profile.

IRI ALGORITHM

Equations 1 through 7 define the IRI. The quarter-car dynamics and the initialization method together make up a type of calculation whose generic name is "*the initial value problem*," or "*integration of ordinary differential equations*." Equations in this form can be solved by several methods. Thus, the IRI can be calculated by more than one method. In this section, two methods are presented and compared.

Euler Integration (Not Recommended)

The most common way to solve ordinary differential equations is by numerical integration. There are many numerical integration algorithms (*10*). The simplest, known as Euler integration, applies the approximation

$$\mathbf{x}_i \approx \mathbf{x}_{i-1} + dt\,\dot{x}_{i-1} \qquad (8)$$

where *dt* is a small time interval proportional to the sample interval

$$dt = \Delta/V \qquad (9)$$

IRI can be calculated using the Euler approximation if the interval between profile samples is sufficiently small; Figure 3 addresses the question of what is sufficiently small? The figure, which shows the output of the algorithm as a function of wavenumber (wavenumber = 1/wavelength), indicates that the algorithm is in error even for sample intervals as small as 10 mm. As the intervals grow to 100 mm (4 in.), the errors become worse. If one is committed to using Euler integration, it is recommended to generate more profile points by linear interpolation to obtain a sample interval of 10 mm or less. The need for a very small interval requires the computer program to perform the calculations many times between the profile measurements, seriously reducing the overall calculation speed.

Recommended Algorithm

Theory

For a set of linear equations such as equation 4, the total response at point *i* is the sum of the *free response* (no input) of the system to



FIGURE 3 Euler integration response plots.

its state at a previous point i-1, plus the *forced response* to an input over the interval between points i-1 and i. In the case that the input is a constant, the closed-form solution is known:

$$\mathbf{x}_i = e^{\mathbf{A}\Delta/V}\,\mathbf{x}_{i-1} + \mathbf{A}^{-1}\,(e^{\mathbf{A}\Delta/V} - \mathbf{I})\,\mathbf{B}u \tag{10}$$

where

e = base of natural logarithms,
$\mathbf{I}$ = 4 × 4 identity matrix, and
u = input that is constant over interval $i - 1$ to i.

The term $e^{\mathbf{A}\Delta/V}$ is a 4 × 4 *state transition* matrix that defines the free response as a linear combination of the four variables in $\mathbf{x}$ at point $i - 1$, and term $\mathbf{A}^{-1}\,(e^{\mathbf{A}\Delta/V} - \mathbf{I})\,\mathbf{B}$ is a four-element *partial response* array that defines the forced response as a linear function of u. The exponential of a matrix can be calculated several ways, one of which is a Taylor series expansion:

$$e^{\mathbf{A}\,\Delta/V} = \mathbf{I} + \sum_{i=i}^{N} \frac{\mathbf{A}^i(\Delta/V)^i}{i!} \tag{11}$$

where $\mathbf{N}$ is a number large enough that the elements of the state transition matrix are correct to within the precision of the computer.

Equation 10 is exact to the extent that the input u is actually constant over the interval from point $i - 1$ to i. Recall that research of different interpolation methods showed that the best approximation of the sampled profile to a continuous one is that the profile slope is constant between samples. This means that to obtain the best accuracy, the assumed constant input u in equation 10 should be profile slope. Then, equation 10 is the solution for the differential equations

$$\dot{\mathbf{x}} = \mathbf{A}\,\mathbf{x} + \mathbf{B}\,s_{ps} \tag{12}$$

where s_{ps} is smoothed profile slope. However, replacing h_{ps} with s_{ps} implies that the array $\mathbf{x}$ is redefined as

$$\mathbf{x} = [s_s, \dot{s}_s, s_u, \dot{s}_u]^T \tag{13}$$

where s_s and s_u are filtered slope variables associated with the sprung and unsprung masses.

Although the state variables in equation 13 are different from those first presented in equation 5, they are also suitable for defining IRI. Specifically, equation 7 can be transformed to

$$\text{IRI} = \frac{1}{n}\sum_{i=i}^{n}\left|s_{s,i} - s_{u,i}\right| \tag{14}$$

The smoothed profile slope, based on a 250-mm interval, is computed with the simple difference equation

$$s_{ps,i} = \frac{h_{p,i+k} - h_{p,i}}{k\Delta} \tag{15}$$

where k was defined in equation 2.

The IRI definition of equation 14 is generalized in the computer code as

$$\text{IRI} = \frac{1}{n}\sum_{i=i}^{n}\left|\mathbf{Cx}\right| \tag{16}$$

where matrix C is defined as:

$$\mathbf{C} = [1\;0\;-1\;0] \tag{17}$$

To initialize the algorithm, the elements of the $\mathbf{x}$ array for $i = 1$ are set as

$$\mathbf{x}_1 = [(h_{p,L_o/\Delta} - h_{p,1})/L_o,\ 0,\ (h_{p,L_o/\Delta} - h_{p,1})/L_o,\ 0] \tag{18}$$

where L_o is 11 m.

Performance

Figure 4 shows the response of the recommended algorithm for several sample intervals. The plots show the state transition method to be much less sensitive to changes in sample interval than the Euler integration method. (Note that the intervals used in Figure 4 are much larger than those shown for the Euler integration method in Figure 3.)



FIGURE 4 Response of recommended algorithm.



FIGURE 5 Error in IRI due to sample interval for two algorithms.

Figure 5 shows theoretical relations between sample interval and IRI for a road with a typical distribution of roughness over wavenumber. It shows that noticeable error exists when the Euler integration is used, even for an interval as small as Δ = 10 mm. As the step grows to 100 mm, the error is about 18 percent. In contrast, the state transition method has negligible error until the sample interval reaches 200 mm. The error remains less than 2 percent at Δ = 300 mm, supporting the original requirement of a 300-mm interval (or smaller) for accurate IRI calculations. However, the original specification of 600 mm for Class 2 measures is associated with a theoretical bias of more than 10 percent, which is probably too much error for most applications.

The exact relationship between sample interval and IRI is specific to a particular profile. The plot in Figure 5 applies to profiles with a typical distribution. A road with proportionally more long-wavelength roughness could be measured with less error than indicated, whereas higher errors would be obtained for a road with proportionally more short-wavelength roughness.

Computer Listings

Figures 6–9 show tested computer routines for calculating IRI, programmed in the Fortran language. The algorithms are simple enough that translation to other languages (e.g., C) should not be too difficult.

Subroutine IRI (Figure 6) should be applied to a profile represented as a one-dimensional array of floating-point numbers. In addition to computing a summary IRI value, the subroutine replaces the original profile ($h_{p,i}$) with a profile filtered with the IRI moving average and Golden Car. This routine contains all details that are highly specific to the IRI, such as (*a*) the specific numerical values of the *Golden Car* parameters, (*b*) the simultaneous conversion of profile elevation to profile slope and smoothing via equation 15, (*c*) the initialization of the filter with the average profile slope over the first 11 m, and (*d*) the accumulation of absolute filtered slope to obtain IRI.

The subroutines in Figures 7–9 are required by the IRI routine; they can also be used for profile analysis applications other than IRI, for filters represented by four differential equations (e.g., other quarter-car models, Butterworth filters, etc.). Subroutine SETABC (Figure 7) sets up the A, B, and C matrices based on quarter-car parameters. (See equations 5 and 17.) Subroutine SETSTM (Figure 8) computes the state transition (ST) and partial response (PR) matrices, given the A and B matrices and a time step DT. It requires a routine INVERT for inverting a square matrix. The code for INVERT is not shown to conserve space, given that suitable matrix algorithms are readily available (*10*). Subroutine STFILT (Figure 9) filters the profile using the state transition method.

The listed code was written for generality and clarity rather than being optimized for efficiency. The algorithm itself is fast enough that coding efficiency is usually not an issue with modern microcomputers. Fully operational profile analysis software is available at no cost from the University of Michigan Transportation Institute (UMTRI) including the source files shown in the listings. Contact the author for information on acquiring the software and source code.

PROFILE MEASUREMENT

The IRI definition describes a method for computing a roughness index for a single longitudinal profile of arbitrary length. Its accuracy and relevance are limited by the quality of the profile measurement, which depends on (*a*) the design and quality of the equipment, and (*b*) the methodology and care used to make the measurement. Following are some ways that profile measurement affects IRI.

Location and Width of Profile

Recall that IRI is defined as a property of single longitudinal profile, measured along a single line down the road. Users often want an overall IRI for a traveled lane. This raises two questions: How many profiles should be taken for a traveled lane? and At what lateral position(s) should the longitudinal profile(s) be measured? The IRI definition does not address this issue, although the recommendation is that the profiles should be measured in two traveled wheel tracks, with the IRI values for each being averaged to obtain a sum-

```
C=====================================================================
      SUBROUTINE IRI(PROF, NSAMP, DX, BASE, UNITSC, AVEIRI)
C=====================================================================
C Filter a longitudinal road profile and calculate IRI.
C
C <-> PROF   REAL      On input, an array of profile height values.
C                      On output, an array of filtered profile values.
C <-> NSAMP  INTEGER   Number of data samples in array PROF. The filtered
C                      profile has fewer points than the original.
C --> DX     REAL      Distance step between profile points (m).
C --> BASE   REAL      Distance covered by moving average (m).
C                      Use 0.250 for unfiltered profile input, and 0.0
C                      for pre-smoothed profiles (e.g. K.J. Law data).
C --> UNITSC REAL      Product of two scale factors: (1) meters per unit
C                      of profile height, and (2) IRI units of slope.
C                      Ex: height is inches, slope will be in/mi.
C                          UNITSC = (.0254 m/in)*(63360 in/mi) = 1069.34
C <-- AVEIRI REAL      The average IRI for the entire profile.

      INTEGER   I, I11, IBASE, J, NSAMP
      REAL      AMAT, AVEIRI, BASE, BMAT, CMAT, DX, PR, PROF, SFPI, ST
      REAL      UNITSC, V, XIN
      DIMENSION AMAT(4, 4), BMAT(4), CMAT(4), PR(4), PROF(NSAMP),
     &          ST(4,4), XIN(4)

C  Set parameters and arrays.
      CALL SETABC(653.0, 63.3, 6.0, 0.15, AMAT, BMAT, CMAT)
      CALL SETSTM(DX/(80./3.6), AMAT, BMAT, ST, PR)
      IBASE = MAX(INT(BASE/DX + 0.5), 1)
      SFPI = UNITSC/(DX*IBASE)

C  Initialize simulation variables based on profile start.
      I11 = MIN(INT(11./DX + .5) + 1, NSAMP)
      XIN(1) = UNITSC*(PROF(I11) - PROF(1))/(DX*I11)
      XIN(2) = 0.0
      XIN(3) = XIN(1)
      XIN(4) = 0.0

C  Convert to averaged slope profile, with IRI units.
      NSAMP = NSAMP - IBASE
      DO 10 I = 1, NSAMP
   10   PROF(I) = SFPI*(PROF(I + IBASE) - PROF(I))

C  Filter profile.
      CALL STFILT(PROF, NSAMP, ST, PR, CMAT, XIN)

C  Compute IRI from filtered profile.
      AVEIRI = 0.0
      DO 20 I = 1, NSAMP
   20   AVEIRI = AVEIRI + ABS(PROF(I))
      AVEIRI = AVEIRI/NSAMP

      RETURN
      END
```

FIGURE 6 Fortran code to calculate IRI from profile.

```
C=======================================================================
      SUBROUTINE SETABC(K1, K2, C, MU, AMAT, BMAT, CMAT)
C=======================================================================
C Set the A, B and C matrices for the a 1/4 car model.
C
C -->  K1    REAL   Kt/Ms = normalized tire spring rate (1/s/s)
C -->  K2    REAL   Ks/Ms = normalized suspension spring rate (1/s/s)
C -->  C     REAL   Ks/Ms = normalized suspension damper rate (1/s)
C -->  MU    REAL   Ks/Ms = normalized unsprung mass (-)
C <--  AMAT  REAL   The 4x4 A matrix.
C <--  BMAT  REAL   The 4x1 B matrix.
C <--  CMAT  REAL   The 4x1 C matrix.

      INTEGER       I, J
      REAL          AMAT, BMAT, CMAT, K1, K2, C, MU
      DIMENSION     AMAT(4, 4), BMAT(4), CMAT(4)

C   Set default for all matrix elements to zero.
      DO 10 J = 1, 4
        BMAT(J) = 0
        CMAT(J) = 0
        DO 10 I = 1, 4
   10     AMAT(I, J) = 0

C   Put 1/4 car model parameters into the A Matrix.
      AMAT(1, 2) = 1.
      AMAT(3, 4) = 1.
      AMAT(2, 1) = -K2
      AMAT(2, 2) = -C
      AMAT(2, 3) = K2
      AMAT(2, 4) = C
      AMAT(4, 1) = K2/MU
      AMAT(4, 2) = C/MU
      AMAT(4, 3) = -(K1 + K2)/MU
      AMAT(4, 4) = -C/MU

C   Set the B matrix for road input through tire spring.
      BMAT(4) = K1/MU

C   Set the C matrix to use suspension motion as output.
      CMAT(1) = -1
      CMAT(3) = 1

      RETURN
      END
```

FIGURE 7 Code to set model matrices.

mary IRI for the lane. Obviously, agreement between measures obtained for a given road by different users is limited unless they choose the same profile locations.

A related issue is the width of the profile. Laser-based systems measure a profile that corresponds to a line several millimeters wide. K.J. Law optical systems measure a line that is about 6 in. wide. DipStick measurements detect the profile under pads that have a fixed diameter. To this author's knowledge, research has not been done to the determine the effect of profile width on the IRI or any other roughness statistic. (And, by the way, the IRI does not specify a profile width.)

Length of Profile

The test sites used in developing IRI all had a length of 320 m (about 0.2 mi). In more recent research studies, 0.1-mi sites are common. Theoretically, IRI can be computed for any length of profile. However, users must realize that the variation in IRI down the road depends on the length over which it is accumulated. When IRI is summarized for 1-mi sections in routine survey work, the highest values (roughest sections) are not as high as when IRI is summarized for 0.1-mi sections. Short distances (e.g., 50 ft) isolate local effects such as faults and produce very high IRI values (*11*).

```
C=========================================================================
      SUBROUTINE SETSTM(DT, A, B, ST, PR)
C=========================================================================
C Compute ST and PR arrays. This requires INVERT for matrix inversion.
C
C -->  DT    REAL   Time step (sec)
C -->  A     REAL   The 4x4 A matrix.
C -->  B     REAL   The 4x1 B matrix.
C <--  ST    REAL   4x4 state transition matrix.
C <--  PR    REAL   4x1 partial response vector.

      INTEGER   I, ITER, J, K
      LOGICAL   MORE
      REAL      A, A1, A2, B, DT, PR, ST, TEMP
      DIMENSION A(4, 4), A1(4, 4), A2(4, 4), B(4), PR(4), ST(4, 4),
     &          TEMP(4, 4)

      DO 20 J = 1, 4
        DO 10 I = 1, 4
          A1(I, J) = 0
   10     ST(I, J) = 0
        A1(J, J) = 1.
   20   ST(J, J) = 1.

C  Calculate the state transition matrix ST = exp(dt*A) with a Taylor
C  series. A1 is the previous term in the series, A2 is the next one.
      ITER = 0
   30 ITER = ITER + 1
      MORE = .FALSE.
      DO 40 J = 1, 4
        DO 40 I = 1, 4
          A2(I, J) = 0
          DO 40 K = 1, 4
   40       A2(I, J) = A2(I, J) + A1(I, K) * A(K, J)

      DO 50 J = 1, 4
        DO 50 I = 1, 4
          A1(I, J) = A2(I, J)*DT/ITER
          IF (ST(I, J) + A1(I, J) .NE. ST(I, J)) MORE = .TRUE.
   50     ST(I, J) = ST(I, J) + A1(I, J)
      IF (MORE) GO TO 30

C  Calculate particular response matrix: PR = A**-1*(ST-I)*B
      CALL INVERT(A, 4)
      DO 60 I = 1, 4
        PR(I) = 0.0
        DO 60 K = 1, 4
   60     PR(I) = PR(I) - A(I, K)*B(K)
      DO 90 J = 1, 4
        DO 70 I = 1, 4
          TEMP(J, I) = 0.0
          DO 70 K = 1, 4
   70       TEMP(J, I) = TEMP(J, I) + A(J, K)*ST(K, I)
        DO 80 K = 1, 4
   80     PR(J) = PR(J) + TEMP(J, K)*B(K)
   90   CONTINUE
      RETURN
      END
```

FIGURE 8 **Code to set state transition matrix.**

```
C=========================================================================
      SUBROUTINE STFILT(PROF, NSAMP, ST, PR, C, XIN)
C=========================================================================
C Filter profile using matrices ST, PR, and C.
C
C <->  PROF    REAL       Input profile. Replaced by the output.
C -->  NSAMP   INTEGER    Number of data values in array PROF.
C -->  ST      REAL       4x4 state transition matrix.
C -->  PR      REAL       4x1 partial response vector.
C -->  C       REAL       4x1 output definition vector.
C -->  XIN     REAL       Initial values of filter variables.

      INTEGER    I, J, K, NSAMP
      REAL       C, PR, PROF, ST, X, XIN, XN
      DIMENSION C(4), PR(4), PROF(NSAMP), ST(4, 4), X(4), XIN(4), XN(4)

C  Initialize simulation variables.
      DO 10 I = 1, 4
   10   X(I) = XIN(I)

C  Filter profile using the state transition algorithm.
      DO 40 I = 1, NSAMP
        DO 20 J = 1, 4
          XN(J) = PR(J)*PROF(I)
          DO 20 K = 1, 4
   20       XN(J) = XN(J) + X(K)*ST(J, K)
        DO 30 J = 1, 4
   30     X(J) = XN(J)
        PROF(I) = X(1)*C(1) + X(2)*C(2) + X(3)*C(3) + X(4)*C(4)
   40   CONTINUE
      RETURN
      END
```

FIGURE 9 Code to filter profile.

It is essential that planners and engineers using IRI understand this fundamental relationship between roughness variation along the road and the length of the road over which the roughness is averaged. Local uses of IRI for pavement management and evaluation of newly constructed pavements should specify a standard length, especially when IRI is used to identify maximum roughness.

Cracks

Profilers with laser sensors can detect cracks, tar strips, patches, and other small bumps. Short events such as these appear as *spikes* in the profile, which partially reduces the effect of such spikes through the 250-mm moving average. However, their influence is not completely eliminated. Consequently, laser-based systems sometimes produce higher IRI measures than other systems.

Bumps are highly objectionable to road users, while cracks go unnoticed. A profiler that can detect bumps should obtain a more accurate and representative IRI value than a profiler that cannot. On the other hand, a profiler that is sensitive to cracks is subject to at least two error sources. First, the extra IRI roughness due to the crack is not indicative of roughness as perceived by the traveling public. Second, the effect is not completely repeatable. Although a system might detect the same crack in repeat measures, the depth of the crack depends on many variables, including the amount of dirt in the crack and the precise lateral location of the laser. In one pass, the crack might appear to be 5 mm deep; in a subsequent pass, it might be seen as 50 mm deep.

Most profilers store data at intervals of 100 mm or more. Once the information is recorded, it is not possible to tell the difference between a crack that is several millimeters wide and a depression that is twice the sample interval (e.g., 200 mm in length). Therefore, it is usually not possible to identify cracks after the measurement is made. The recommended solution is for developers of laser-based profilers to install real-time crack detection software. If enabled (for roughness measurement), the software should hold the previous value when the laser goes into a crack. The algorithm must distinguish between bumps, which cause roughness, and cracks, which do not. Linear filtering does not distinguish between bumps and cracks and therefore is not sufficient.

The problem of cracks is not unique to IRI. It exists with almost any roughness index.

Reference Profilers

World Bank Report 46 defined two classes of profiling methods that were later adopted by FHWA for the HPMS data base (*7,8*). Profilers are considered Class 2 if they produce IRI measures that are neither high nor low on the average. However, an individual measurement is expected to have random error. Some profilers clearly are more accurate than others, so the concept of a Class 1 measurement was introduced to define a reference that can be used to determine the accuracy of other instruments. A Class 1 instrument must be so accurate that the random error is negligible: its IRI measure is "the truth."

On the basis of data available at the time of the IRRE, the level of accuracy associated with Class 1 was set at ±2 percent for 320-m (0.2-mi) test sites. Computer studies of the sensitivity of IRI to sample interval and height resolution were used to define a Class 1 profiling instrument. The concept of classes for profiler methods has proved popular among users and manufacturers. For example, ASTM recently approved a standard (not yet published) on inertial profilers that establishes four classes of profilers. However, evidence from correlation studies over the past 10 years indicates that current specifications of a Class 1 IRI profile measurement are not sufficient. Devices that on paper qualify as Class 1 do not always show the repeatability that was expected. In retrospect, the major problem is that the specifications focus on the equipment and not its use. Even with highly experienced operators, human error sometimes results in an incorrect profile measurement, thus an incorrect IRI. (For example, operators can start the measurement too soon or too late, or locate the instrument in the wrong lateral position.)

Some limitations with the equipment are not covered by the specification of sample interval and height resolution. Experience now shows that a device might have trouble with a specific road surface. A common example is that certain textured surfaces can confuse even the best noncontacting height sensors used in high-speed profilers. Another example is that the DipStick, which captures samples of profile height every 300 mm, can miss significant roughness on a surface with bumps just several inches long.

Hardware specifications can qualify a device as Class 1–capable, but, as a minimum, repeat measures are needed to qualify the measures taken on a specific site. Repeatability is better with static devices such as the DipStick. However, good repeatability does not always imply good accuracy. If the profile measurement is always started at the same point (with a tolerance of an inch or so), the IRI can be highly repeatable but incorrect if the sample interval (Δ) is not sufficient to capture all significant roughness. Thus, for static measures, repeat measures should be made at slightly different starting locations.

As experience is gained, areas emerge where differences between instruments are not easily explained. For example, it is known that profiles of some roads change significantly with temperature. Provisions must be made to account for this effect in the acquisition of profile data for research involving Class 1 methods.

Research is needed to refine the definition of a Class 1 measurement. As a minimum, the specification should be extended to describe a test method for using the instrument that allows the user to estimate its quality (e.g., by looking at the scatter in repeated measures).

CONCLUSION

When the IRI was defined in World Bank Technical Report 46, there were only about a half-dozen inertial profilometers in America. Since then profiling has become the primary means for measuring road roughness in the United States. More than half the states have purchased or built profiling systems. The federal government maintains a fleet of profilers for calibration and research programs, and consulting companies maintain profiling systems to provide measures to states and local districts that do not have their own equipment. FHWA has encouraged profiler use and has sponsored several correlation experiments. Profiler users have organized into the Road Profiler User Group, which has established an annual correlation experiment for several years in which users are invited to measure profiles and IRI for test sites.

The profilers in use cover a wide variety of sensor types, cost, and analysis options. Limited by the speed of sound, systems with ultrasonic sensors can measure profile at intervals no closer than about 300 mm (1 ft) at highway speeds. Other systems, with laser sensors, can measure at intervals going down to a few millimeters. Some systems perform minimal profile filtering. Others routinely smooth the data to avoid aliasing and remove long wavelengths to standardize plot appearances. Even with these differences, most profilers in use can obtain IRI measures that show reasonable agreement (within 5 percent).

However, recent correlation experiments show that no existing profiler can measure "true IRI" with the high accuracy one might expect of a Class 1 instrument (i.e., within 2 percent). Further research is needed to determine the reasons that consistent measures of roughness are not obtained. Two possible sources of discrepancy are user practice and changes in road profile due to temperature and environmental effects.

ACKNOWLEDGMENTS

Steve Karamihas of UMTRI assisted with the preparation of this paper and provided the theoretical response plots. The preparation of this paper was funded by FHWA.

Elson Spangler of Surface Dynamics and John Darlington of the Michigan Department of Transportation first applied quarter-car simulation to measure road profiles in the early 1970s. Spangler also served on the scientific committee for the work described in *NCHRP Report 228* and was an early advocate of a Golden Car mathematical model. Rudy Hegmon of FHWA also participated in the NCHRP work and has long been an advocate of profile-based measures of road roughness. Hegmon has led FHWA's involvement in profiler technology since the late 1970s; he initiated the development of FHWA's PRORUT system. Thomas Gillespie of UMTRI was the principal investigator for the NCHRP project, technical director of the World Bank Research program in which IRI was developed, and coauthor of the IRI specifications. Cesar Queiroz (now with the World Bank) was the overall director of the World Bank project, which was run by the Brazilian Road Research Institute (IPR/DNER) and the Brazilian Transportation Planning Agency (GEIPOT). William Paterson of the World Bank was an early advocate of the development of an IRI and a coauthor of the IRI specifications. The late Per Fossberg of the World Bank contributed to the final stages of the IRI development and the publication of World Bank Technical Reports 45 and 46. The person most responsible for the existence of the IRI is Clell Harral of the World Bank. Harral conceived the idea of the IRI and arranged for the participation of the various technical personnel who developed the specifications, tested them, and wrote the guidelines for applying them.

REFERENCES

1. Spangler, E. B., and W. J. Kelly. *GMR Road Profilometer, a Method for Measuring Road Profile.* Research Publication GMR-452. General Motors Corp., Warren, Mich. Dec. 1964.
2. Buchanan, J. A., and A. L. Catudal. Standardizable Equipment for Evaluating Road Surface Roughness. *Public Roads*, Feb. 1941.
3. Darlington, J. R. *Evaluation and Application Study of the General Motors Corporation Rapid Travel Profilometer.* Research Report R-731. Michigan Department of State Highways, Lansing, Oct. 1970.
4. Burchett, J. L., et al. *Surface Dynamics Profilometer and Quarter-Car Simulator: Description, Evaluation, and Adaptation.* Research Report 465. Kentucky Department of Transportation, Frankfort, 1977.
5. Gillespie, T. D., M. W. Sayers, and L. Segel. *NCHRP Report 228: Calibration of Response-Type Road Roughness Measuring Systems.* TRB, National Research Council, Washington, D.C., Dec. 1980.
6. Sayers, M. W., T. D. Gillespie and C. Queiroz. *International Experiment to Establish Correlations and Standard Calibration Methods for Road Roughness Measurements.* World Bank Technical Paper 45. World Bank, Washington, D.C., 1986.
7. Sayers, M. W., T. D. Gillespie, and W. D. Paterson. *Guidelines for the Conduct and Calibration of Road Roughness Measurements.* World Bank Technical Paper 46. World Bank, Washington, D.C., 1986.
8. *Highway Performance Monitoring System, Field Manual, Appendix J.* Order M 5600.1A. FHWA, U.S. Department of Transportation, April 1990.
9. Sayers, M. W. Two Quarter-Car Models for Defining Road Roughness: IRI and HRI. In *Transportation Research Record 1215*, TRB, National Research Council, Washington, D.C., 1989, pp. 165–172.
10. Press, W., et al. *Numerical Recipes: The Art of Scientific Computing.* Cambridge University Press, England, 1986.
11. Sayers, M. W. Profiles of Roughness. In *Transportation Research Record 1260*, TRB, National Research Council, Washington, D.C., 1990, pp. 106–111.

Publication of this paper sponsored by Committee on Surface Properties–Vehicle Interaction.